
	Function name
	f_TC_10_1_2_1_NR5GC

	Reason for change
	The TTCN implementation for the preamble calls the function f_NR5GC_Preamble (v_NGC_NASCellA, STATE_CONNECTED_3A, TEST_LOOPModeB_ON);, which calls Activate RB test Mode, and Close UE Test Loop Mode B.
Depending on whether the first PDU session established is for IMS, or not the TTCN implementation calls :
if (not (isvalue(v_DRBInfo))) {
      f_UT_RequestPDUSession(UT, 2, pc_APN_ID_Internet);
      f_NR5GC_PDUSessionEstablishment (v_NGC_NASCellA, 1);
      f_NR5GC_Activate_TestMode (v_NGC_NASCellA, tsc_UE_TestLoopMode_TypeB, tsc_NR_RbId_SRB2);  // activate - SEE CHANGE 3
      if(not pc_IP_Ping) {
          f_NR5GC_CloseUE_TestLoopModeB (v_NGC_NASCellA, '00'O);
      }
    }

This results in the second call for Activate RB Test Mode, and Close UE Test Loop Mode B.
This second call can be ignored by the UEs, because of the first call and there is no Open UE Test Loop ModeB function call before the second call.

This causes testcase failure at Step 4, because the UE does not loop back the data that is received at Step 3.

This needs to be corrected.

	Summary of change
	Updated the TTCN implementation to ensure that Activate RB Test Mode and Close UE Test Loop is called only once after the non-IMS PDU session is established.

	TTCN module
	GSM_NR5GC.ttcn



Before Change:
	function f_TC_10_1_2_1_NR5GC() runs on NR5GC_PTC
  {
    //Network-requested PDU session modification / Accepted / Reactivation
    var NR_CellId_Type v_NGC_NASCellA := f_NR5GC_MapNASCell(ngc_CellA, NG_AllCellsOnSamePLMN);
    var integer v_DRBId; // @sic R5s230095 sic@
    var template (omit) NR_DRBInfo_Type v_DRBInfo; // @sic R5s230095 sic@

    f_NR5GC_Init_NAS(NR_1);

    //Create and configure cells
    f_NR_CellConfig_Def(v_NGC_NASCellA);
    
    //Preamble
    //The UE is in 5GS state 3N-A
    f_NR5GC_Preamble (v_NGC_NASCellA, STATE_CONNECTED_3A, TEST_LOOPModeB_ON); // @sic R5-198972 sic@

    // check if internet PDU session established in preamble // @sic R5s230095 sic@
    v_DRBId :=  f_NR_GetDefaultDRB_ForFirstNonIMSPDUSession();
    v_DRBInfo := f_NR_GetDRBInfo_FromId(v_DRBId);
    if (not (isvalue(v_DRBInfo))) {
      f_UT_RequestPDUSession(UT, 2, pc_APN_ID_Internet);
      f_NR5GC_PDUSessionEstablishment (v_NGC_NASCellA, 1);
      f_NR5GC_Activate_TestMode (v_NGC_NASCellA, tsc_UE_TestLoopMode_TypeB, tsc_NR_RbId_SRB2);  // activate - SEE CHANGE 3
      if(not pc_IP_Ping) {
          f_NR5GC_CloseUE_TestLoopModeB (v_NGC_NASCellA, '00'O);
      }
    }

    f_NR_TestBody_Set(true); // @sic R5s220796 sic@
    //Main test body
    fl_TC_10_1_2_1_Body();
    f_NR_TestBody_Set(false); // @sic R5s220796 sic@

    // Finish
    //f_NR5GC_OpenUE_TestLoopMode_Deactivate_TestMode(v_NGC_NASCellA); // @sic R5s190944 sic@ //@sic R5s230130 sic@
    f_NR_Postamble(v_NGC_NASCellA, STATE_CONNECTED_3A);
  }


After Change:

	function f_TC_10_1_2_1_NR5GC() runs on NR5GC_PTC
  {
    //Network-requested PDU session modification / Accepted / Reactivation
    var NR_CellId_Type v_NGC_NASCellA := f_NR5GC_MapNASCell(ngc_CellA, NG_AllCellsOnSamePLMN);
    var integer v_DRBId; // @sic R5s230095 sic@
    var template (omit) NR_DRBInfo_Type v_DRBInfo; // @sic R5s230095 sic@

    f_NR5GC_Init_NAS(NR_1);

    //Create and configure cells
    f_NR_CellConfig_Def(v_NGC_NASCellA);
    
    //Preamble
    //The UE is in 5GS state 3N-A    
    f_NR5GC_Preamble (v_NGC_NASCellA, STATE_CONNECTED_3A, TESTModeB_ON); 

    // check if internet PDU session established in preamble // @sic R5s230095 sic@
    v_DRBId :=  f_NR_GetDefaultDRB_ForFirstNonIMSPDUSession();
    v_DRBInfo := f_NR_GetDRBInfo_FromId(v_DRBId);
    if (not (isvalue(v_DRBInfo))) {
      f_UT_RequestPDUSession(UT, 2, pc_APN_ID_Internet);
      f_NR5GC_PDUSessionEstablishment (v_NGC_NASCellA, 1);      
    }  
    
    if(not pc_IP_Ping) { 
          f_NR5GC_CloseUE_TestLoopModeB (v_NGC_NASCellA, '00'O);
    }

    f_NR_TestBody_Set(true); // @sic R5s220796 sic@
    //Main test body
    fl_TC_10_1_2_1_Body();
    f_NR_TestBody_Set(false); // @sic R5s220796 sic@

    // Finish
    //f_NR5GC_OpenUE_TestLoopMode_Deactivate_TestMode(v_NGC_NASCellA); // @sic R5s190944 sic@ //@sic R5s230130 sic@
    f_NR_Postamble(v_NGC_NASCellA, STATE_CONNECTED_3A);
  }



